Calculus WIZ and
The Mathematical Explorer – advanced use

Numerical and symbolical capabilities

Both *Calculus WIZ* and *The Mathematical Explorer* support many symbolic and numeric capabilities of its parent product *Mathematica.* You can use *Mathematica* arbitratry precision arithmetics to enumerate famous mathematical constants to as many decimal places as you wish:

```
N[Pi, 120]
```

```
3.1415926535897932384626433832795028841971693993751058209\
7494459230781640628620899862803482534211706798214808651 3\
28230665
```

We got Ludolf's number to 120 decimal places and we can do the same with Euler's constant:

```
N[E, 120]
```

```
2.7182818284590452353602874713526624977572470936999595749\
66967627724076630353547594571382178525166427427466391932\
00305992
```

You can work with relations between constants symbolically as well:

```
E^(I Pi)
```

```
-1
```

Sometimes you will not get immediate answer such as in this example where we calculate the imaginary unit to the power of itself:

```
I^I
```

i^i

The numerical value of the imaginary unit to the power of itself is

```
N[I^I]
```

```
0.20788 + 0. i
```

The Mathematical Explorer can give symbolic value easily too:

```
ComplexExpand[I^I]
```

$$e^{-\pi/2}$$

This *Mathematica* command is disabled in *Calculus WIZ*, however. If you type

```
ComplexExpand[I^I]
```

into your *Calculus WIZ* session, the response will be following:

```
Disabled`ComplexExpand[i^i]
```

You can easily find a solution to algebraic equation

```
Solve[x^3 == 1, x]
```

$$\{\{x \to 1\}, \{x \to -\sqrt[3]{-1}\}, \{x \to (-1)^{2/3}\}\}$$

and use the reset in subsequent calculations. Note that the symbol % refers to the result of the previous calculation and the symbol /. performs substitution, i.e., x is replaced by the value on the other side of the arrow (x /. x->a produces a) :

```
x /. %
```

$$\{1, -\sqrt[3]{-1}, (-1)^{2/3}\}$$

The resulting complex numbers can be transformed to their algebraic forms:

```
ComplexExpand[%]
```

$$\left\{1, -\frac{1}{2} - \frac{i\sqrt{3}}{2}, -\frac{1}{2} + \frac{i\sqrt{3}}{2}\right\}$$

and/or evaluated numerically:

```
N[%]
```

```
{1., -0.5 - 0.866025 i, -0.5 + 0.866025 i}
```

As an example of more advanced symbolic manipulation, let us mention conversion of trigonometric and hyperbolic functions to exponential function and *vice versa*.

```
TrigToExp[Sin[z]]
```

$$\frac{1}{2} i e^{-iz} - \frac{1}{2} i e^{iz}$$

This transforms complex exponentials to trigonometric and hyperbolic functions:

```
ExpToTrig[%]
```

Sin[z]

The result is not surprising. But the following application is. Let us convert third power of the sine function into exponentials, expand the power by binomical formula and convert the result back into trigonometric functions. Her is what you get:

```
TrigToExp[Sin[z]^3]
```

$$-\frac{1}{8} i \left(e^{-iz} - e^{iz}\right)^3$$

```
Expand[%]
```

$$\frac{3}{8} i e^{-iz} - \frac{3}{8} i e^{iz} - \frac{1}{8} i e^{-3iz} + \frac{1}{8} i e^{3iz}$$

```
ExpToTrig[%]
```

$$\frac{3\,\text{Sin}[z]}{4} - \frac{1}{4}\,\text{Sin}[3z]$$

If you do not believe that

$$\frac{3\,\text{Sin}[z]}{4} - \frac{1}{4}\,\text{Sin}[3z] == \text{Sin}[z]\,\hat{}\,3$$

check some mathematical tables or simply write

$$\text{Simplify}\left[\frac{3\,\text{Sin}[z]}{4} - \frac{1}{4}\,\text{Sin}[3z] == \text{Sin}[z]\,\hat{}\,3\right]$$

You will see that it is

True

The function ExpToTrig is supported in both products and it can be used instead of ComplexExpand function when dealing with expression such as $-\sqrt[3]{-1}$:

```
ExpToTrig[-(-1)^(1/3)]
```

$$-\frac{1}{2}-\frac{i\sqrt{3}}{2}$$

Programmable capabilities of Calculus WIZ and The Mathematical Explorer

As *Mathematica* based products, *Calculus WIZ* and *The Mathematical Explorer* both support procedural, functional (LISP programming style using pure functions) and rule based programming paradigms. For that reason, their capabilities are not limited to predefined templates and they can be used in a creative way by writing the user's own programs.

We can easily generate tables of function values:

```
Table[Sin[x], {x, 0, Pi, Pi/6}]
```

$$\left\{0, \frac{1}{2}, \frac{\sqrt{3}}{2}, 1, \frac{\sqrt{3}}{2}, \frac{1}{2}, 0\right\}$$

Taking advantage of the functional programming style, we can write short and elegant programs. Here we create a table of derivates and integrals for a given list of functions:

```
({#, D[#, x], Integrate[#, x]} & /@ {Sin[x], Cos[x], Tan[x], Cot[x],
ArcTan[x]}) // TableForm
```

Sin[x]	Cos[x]	-Cos[x]
Cos[x]	-Sin[x]	Sin[x]
Tan[x]	$Sec[x]^2$	-Log[Cos[x]]
Cot[x]	$-Csc[x]^2$	Log[Sin[x]]
ArcTan[x]	$\frac{1}{1+x^2}$	$x\,ArcTan[x]-\frac{1}{2}Log[1+x^2]$

It is also easy to create graphs of functions and combine them:

```
Plot[{Sin[x], Cos[x]}, {x, 0, 2 Pi}]
```

Rule based programming can be used to perform e.g. substitutions in equation solving.

```
Sin[x]^2+Cos[x]==1/2  /. Sin[x] -> Sqrt[1-Cos[x]^2]
```

$$1+Cos[x]-Cos[x]^2 == \frac{1}{2}$$

The symbol % refers to result of previous calculation:

```
% /. Cos[x] -> z
```

$$1+z-z^2 == \frac{1}{2}$$

The command Solve solves the previous equation to which we refer by %:

```
Solve[%, z]
```

$$\left\{\left\{z\to\frac{1}{2}\left(1-\sqrt{3}\right)\right\},\left\{z\to\frac{1}{2}\left(1+\sqrt{3}\right)\right\}\right\}$$

Results can immediately be used for back-substitution:

```
Cos[x] == z /. %
```

$$\left\{\text{Cos}[x]==\frac{1}{2}\left(1-\sqrt{3}\right),\text{Cos}[x]==\frac{1}{2}\left(1+\sqrt{3}\right)\right\}$$

and we can solve the first one of these two goniometric equations

```
Solve[%[[1]], x]
```

Solve::ifun Inverse functions are

being used by Solve, so some solutions

may not be found.

$$\left\{\left\{x\to -\text{ArcCos}\left[\frac{1}{2}\left(1-\sqrt{3}\right)\right]\right\},\left\{x\to \text{ArcCos}\left[\frac{1}{2}\left(1-\sqrt{3}\right)\right]\right\}\right\}$$

Note that the warning message was generated as the function is not invertible on the entire domain. Since both *Calculus WIZ* and *The Mathematical Explorer* support computations with exact quantities, the previous output is not written as a number. We can, however, ask them for numeric solutions:

```
N[%]
```

```
{{x → -1.94553}, {x → 1.94553}}
```

Graphical capabilities

Both products support wide range graphical representation of objects: basic graphical shapes (so called graphical primitives), 2D and 3D graphs of functions and contour graphics.

The following command will draw point and line.

```
Show[Graphics[{Line[{{0, 0}, {1, 1}}], PointSize[0.02],
  Point[{1/2, 1}]}]]
```



Case study: Find intersections of the graphs of functions sine of x and second power of x in the first quadrant.

To get an idea how the situation looks like, we plot the graphs of these two functions:

```
Plot[{Sin[x], x^2}, {x, 0, Pi}]
```



Now we can search for the intersection point:

$\text{Solve}[\text{Sin}[x] == x^2, x]$

Solve::tdep : The equations appear to involve the variables to be solved for in an essentially non−algebraic way.

$\text{Solve}[\sin(x) == x^2, x]$

It is impossible to find solution to transcendental equations in closed form in general. Thus we are limited to a search for solutions by some approximative numerical method. To this end, we use command FindRoot:

```
{x, Sin[x]} /. FindRoot[Sin[x] == x^2, {x, 0.1}]
```

$$\{-2.68703\times 10^{-8}, -2.68703\times 10^{-8}\}$$

We were trapped by a trivial solution in this case (the value 0.1 is initial value for x start the search). From the picture above, we see, that the second solution is somewhere between 0.5 and 1. Let us start with the value 0.5:

```
pt = {x, Sin[x]} /. FindRoot[Sin[x] == x^2, {x, 0.5}]
```

{0.876726, 0.768649}

Now we catched the nontrivial solution. We can put it together into one picture:

```
Plot[{Sin[x], x^2}, {x, 0, Pi},
 Epilog :> {PointSize[0.03], Point[{0, 0}], Point[pt]}]
```



Data processing

Due to the fact, that Calculus WIZ and The Mathematical Explorer support basic data processing, their use is not limited to math classes.

Here we produce some random data (in reality these dat will come from measurement):

```
data = Table[{k, k^2 + 2 k + Random[] / 10}, {k, 1, 10}]
```

Note that Calculus WIZ does not support command Random[], so you will need some external real data to perform following calculation.

1	3.07191
2	8.0146
3	15.0765
4	24.034
5	35.0254
6	48.0838
7	63.0064
8	80.0063
9	99.0093
10	120.059

```
ListPlot[data, PlotStyle→ {PointSize[0.03]},
 AxesOrigin→ {0, 0}]
```



This command finds a least-squares fit to a list of data as a linear combination of the functions 1, x and x^2.

```
Fit[data, {1, x, x^2}, x]
```

$$1.00074x^2 + 1.9887x + 0.0725026$$

Let us put data and the graph of theinr fit into one picture:

```
Plot[1.0007380890493205` x^2 + 1.988699387191992` x+
  0.07250258820562902`, {x, 0, 10},
 Epilog:→ {PointSize[0.02], Point /@data}]
```



Finaly let us to note that Calculus WIZ and The Mathematical Explorer do not support import and export of data into files. The data can be exchanged through clipboard, however.

Conclusion

Both *Calculus WIZ* and *The Mathematical Explorer* are stand-alone software products based on solid *Mathematica* technology. They are suitable not only for exploring mathematical ideas by means of templates, but also for using work created by students through programming. Both products support the *Mathematica* programming core language (with some restrictions) with its rule-based, functional and procedural paradigms. Both products also support many *Mathematica* commands for formula manipulation, symbolic/numeric integration etc. They do not support *Mathematica* commands for processing huge data sets as these commands are not supposed to be used by high school students and undergraduate students. There are also some other restrictions with respect to the full version of *Mathematica*. *Calculus WIZ* and *The Mathematical Explorer* are both recommended for later courses at high school and/or for teaching undergraduate courses at university and college level.